
Unit 6: Coordinates   Application: Random Walk   

Overview: Explorations with random numbers can lead 

to fascinating observations. This application gives you 

opportunities to explore more probability oddities and to 

program the Rover to move around on a grid. 

Objectives: 

 Use coordinate movement to simulate a ‘random walk’ 

 Use counters and accumulators in a program 

 Use compound conditions with not and and 

 

A ‘random walk’ is a computer programming experiment. This activity ties together several different programming skills. 

The Problem: 

Suppose your town’s streets are laid out in a square grid pattern and that your 

school is located at (0,0) on the grid. Your home is at (7,3), which represents 7 

blocks east and 3 blocks north of the school. (Consider changes to this position.) 

The image to the right is a Stat Plot of these two points. 

Starting at the school, you walk one block in a random direction (north, south, east, 

or west). At each intersection, you then walk one block in random direction again. 

Will you ever make it home? How many blocks will you walk before getting home? 

Will you quit from exhaustion before you get home?  

 

Planning is an important part of coding. Think about what the Rover can do and what your programming language can do. 

Bear in mind that, when working with random numbers, we are at the mercy of the machine. It may take a long time for 

the Rover to reach home, so we’ll add a stipulation that the Rover is allowed only a limited number of blocks before 

quitting. 

Teacher Tip: Prepare students to allow the program to Input the location of the home, 

(7,3). Use variables for this position from the start. In fact, it is better to use variables as 

much as possible (rather than specific numbers). The use of variables allows one to easily 

change values or modify the program to use Input statements rather than Store statements 

to initialize the state of the simulation. 

1. Begin this program with the CONNECT RV statement. Set the Rover’s grid 

size to 5 cm using Send(“SET RV.GRID.M/UNIT .05”).  

Recall that this command is found in the prgm > Hub > Rover (RV)… > RV 

Setup… menu and it changes the Rover’s ‘unit’ of movement (used in 

FORWARD 1) from 10 cm to 5 cm, allowing for more grid points in a smaller 

space. 

Initializing Variables 

2. Store the home location (7,3) and the starting number of blocks walked (0) in 

the variables A, B and W, respectively. The variable W will be used to keep 

track of the number of blocks the Rover walks and will be used to stop the 

program if the Rover walks too far. The Rover quits from exhaustion when it 

walks a specified number of blocks.  

Let’s use 10 blocks walked as a value for the exhaustion distance. Use the 

variable E for this value: 10E.  

3. The Rover arrives home when its position, which we refer to as (X, Y), is     

(7, 3). Initialize both of the variables, X and Y, to be zero. 

 

 
The Main Loop 

4. The main loop of the program consists of a While loop with two conditions to 

be addressed. The conditions are making it home (when X=A and Y=B) or 

quitting from exhaustion (when W = E).  The Rover quits when the variable 

W (the number of blocks walked) is equal to the value stored in E. 

 

The While loop continues as long as those conditions are false, so we set up 

the opposite conditions. 

 

The While loop will be:  

While W<E and not(X=A and Y=B) 

Remember that fixed values related to the problem are stored in E, A, and B.  

not( ) is used to ensure that the program continues as long as the Rover is 

not ‘home’. Logically, and not( is the ‘opposite’ of or. 

5. Remember to include an End statement for the loop. 

 

Inside the loop 

6. Increment W (the number of blocks walked): W+1W. 

 

7. Pick a random direction (north, south, east, or west): 

o The Rover’s TO ANGLE command turns the Rover to an ‘absolute’ direction: 0 is east, 90 is north, 180 is 

west, and 270 is south 

o randInt(0,3) gives a random Integer from the set 0, 1, 2, and 3 

o Multiply that value by 90 to get 0, 90, 180, or 270 

o The statement to pick a random Direction D is: 

90randInt(0,3)D 

8. Turn the Rover to the Angle D: Send(“RV TO ANGLE eval(D)”).  

 

9. Move the Rover forward 1 unit (1 block in our simulation): Send(“RV FORWARD 1”). 

 

10. Update the Rover’s position in the program: 

o If the Rover goes north, then increase Y by 1 

o If the Rover goes east, then increase X by 1 

o If the Rover goes south, then decrease Y by 1 

o If the Rover goes west, then decrease X by 1 

Include some Wait commands to keep the program in sync with the Rover’s movements. Turning takes time and moving 

forward takes time. The Wait times depend on the angle of the turn and the distance travelled so you may need to 

experiment with the Wait values.  

After the Loop 

The loop ends in one of two ways: 

 If W=E, then the Rover quits walking from exhaustion. Play a sad song, display a red color on the Rover’s LED, 

and display “ROVER QUIT” on the calculator screen. 

 If X=A and Y=B, the Rover made it home. Play a happy song, display a green color on the LED, and display a 

“ROVER MADE IT HOME” message on the calculator screen.  
 

Can you get the Rover to do a ‘happy dance’ when it arrives home? 

Teacher Tip:  

Sample Solution: Students can test their code without a Rover. The Send commands will 

not do any harm. The first Disp statement in the code below tells where the Rover is on its 

coordinate grid even if the Rover is not available. The Wait time may need adjustment and 

can be removed to run the program faster without the Rover but they help keep the 

calculator in sync with the Rover. Also feel free to modify the total number of blocks (10) 

and the target point’s coordinates (7,3). Sample Solution: 

ClrHome 

Disp "ROVER UNIT6 APP" 

Send("CONNECT RV") 

Send("SET RV.GRID.M/UNIT .05") 

Pause "PRESS ENTER TO START" 

7→A 

3→B 

0→W 

10→E 

0→X 

0→Y 

While W<E and not(X=A and Y=B) 

W+1→W 

90randInt(0,3)→D 

Send("RV TO ANGLE eval(D)") 

Send("RV FORWARD 1") 

Wait 2 

If D=0:X+1→X 

If D=90:Y+1→Y 

If D=180:X-1→X 

If D=270:Y-1→Y 

Disp {X,Y} 

End 

If W=E:Disp "ROVER QUIT" 

If X=A and Y=B:Disp "ROVER MADE IT HOME" 

 



 
10 Minutes of Code    UNIT 6: APPLICATION 

             TI-84 PLUS CE WITH THE TI-INNOVATOR™ ROVER TEACHER NOTES 

©2018 Texas Instruments Incorporated 4 education.ti.com 

Possible Extensions:  

 Light up the COLOR LED in a different color for each direction moved (and only while 

moving - not turning). Timing would be critical. 

 Add SOUND for special effects. 

 Add a test to see if the Rover is ever back at school after it starts to move. This could be 

another condition for ending the program that could be added to the While condition 

(but not at the beginning since the loop would never get started). 

 Add code to keep track of and display the total number of blocks the Rover walked and 

the final (as the crow flies) distance (displacement) from the starting point. 

 Consider limiting the Rover to only walk east or north. How does this affect the logic of 

the program? (Hint: See the While condition.) 

Bonus 1: How far from the school (as the crow flies) is the Rover at the end of the program?  

Bonus 2: What was the maximum distance from the school? From home? 


